

UNITED STATES DEPARTMENT OF THE INTERIOR
GEOLOGICAL SURVEY

Coal-seismic, desktop computer programs in BASIC;
part 5: perform X-square/T-square analyses and
plot normal moveout lines on a seismogram overlay

by

1

Wilfred P. Hasbrouck

Open-File Report 83-350

This report is preliminary and has not been reviewed for conformity with U.S. Geological Survey editorial standards. Any use of trade names is for descriptive purposes only and does not imply endorsement by the USGS. Although the computer programs in this report have been extensively tested, the USGS makes no guarantee whatever of correct results.

1

Golden, Colorado

1983

ABSTRACT

Processing of data taken with the U.S. Geological Survey's coal-seismic system is done with a desktop, stand-alone computer. Programs for this computer are written in the extended BASIC language used by the Tektronix 4051 Graphic System. This report presents computer programs to perform X-square/T-square analyses and to plot normal moveout lines on a seismogram overlay.

INTRODUCTION

Since the early days of reflection-seismic exploration, plots have been made of the shotpoint-to-detector distance (X) squared versus reflection arrival time (T) squared in order to obtain a velocity, and consequently a depth, to a specified reflection horizon (Green, 1938). For the ideal case of a single layer of constant velocity, the inverse slope of an X-square (independent variable)/T-square (dependent variable) plot exactly equals the square of the constant (average) velocity of this layer. When the model consists of a set of parallel layers in which the layer velocities (the interval velocities) are constant, Dix (1955) has shown that the square of the inverse slope is the root-mean-square (rms) velocity (Telford and others, 1976, p. 271). Full discussion of that quantity with the units of reciprocal velocity squared represented by the slope of the least-square straight-line on an X-square/T-square plot is beyond the scope of this report. Let us simply call the velocity so obtained the X-square/T-square velocity.

The purpose of the first section of this report is to discuss a computer program used to determine a velocity from an X-square/-Tsquare plot and to list the negative of the deviations from the least-square, straight line drawn through these data. For a group of reflections recorded on a single seismic record, these listed values constitute a statistical set of static corrections. If the reflection times have been corrected by first-arrival procedures prior to entry into the X-square/T-square program, then the deviations represent additional static shifts beyond the reach of near-surface refraction arrivals.

The second program of this report presents a procedure to plot normal-moveout lines on a seismic-record overlay. In working with seismic records targeted at shallow horizons, use of these overlays can help the interpreter isolate those arrivals that have a higher likelihood of being reflections.

These data processing procedures were developed as part of the U.S. Geological Survey's coal-seismic system. All computer programs were written in an extended BASIC language developed by Tektronix, Inc. for use with their 4051 Graphic System. Two pieces of Tektronix computing equipment are required by the programs: a 4051 Graphic System with a 32K-byte memory and a 4631 Hard Copy unit.

All programs are self-prompting. In tracing through a sample problem, you will notice that the programs print questions and requests followed by a flashing question mark. The computer then waits for you to enter a response from the keyboard. Replies entered in order to run the sample problems are enclosed in boxes on the figures of this report.

X-SQUARE/T-SQUARE ANALYSES

In this report, X-square/T-square analyses are used to determine a velocity to a selected reflecting horizon and to compute and list the negative of the deviations from the least-square straight line through the X-square/T-square plot. To a first approximation, these tabulated values can be thought of as representing residual statics--those that remain after the static corrections obtained from first-arrival procedures have been applied. Caution!, blind application of these residual statics can be a dangerous procedure when only a few good reflections, such as those from a coal horizon, are present on the records because you may inadvertently remove the very structure on the coal horizons that you seeked to find by running the seismic survey.

Figure 1 is a copy of the screen display showing the responses needed in order to produce the results shown on figure 2. All distances are in meters and all times are in milliseconds (msec). Data used in this example were taken as part of a shear-wave experiment conducted in an area covered with glacial drift (Hasbrouck and Padgett, 1982).

X-SQUARE, T-SQUARE ANALYSIS PROGRAM

HEADER ON RECORD (8 characters)	=	21041015
AREA NAME (20 characters,max)	=	THORNE COLLIERY
LATITUDE (12 characters,max)	=	53 38 15.5 N
LONGITUDE (13 characters,max)	=	00 56 02 W
SOURCE TYPE (18 characters,max)	=	HAMMER SHEAR TO SW
DETECTOR TYPE (18 characters,max)	=	TRANSVERSE, 40 HZ
DISTANCE TO FIRST DETECTOR	=	48
DISTANCE TO LAST DETECTOR	=	135
DETECTOR INTERVAL	=	3

INDEX NO.	DETECTOR DISTANCE	REFLECTION TIME
1	48	428
2	51	434
3	54	440
4	57	443
5	60	450
6	63	457
7	66	466
8	69	477
9	72	485
10	75	492
11	78	500
12	81	507
13	84	519
14	87	531
15	90	541
16	93	554
17	96	563
18	99	573
19	102	584
20	105	594
21	108	604
22	111	614
23	114	626
24	117	642
25	120	653
26	123	669
27	126	681
28	129	693
29	132	707
30	135	717

Figure 1. Copy of screen display showing sample of data entries for the X-square/T-square program.




Figure 2. Copy of screen display showing the results produced by the X-square/T-square program using information entered on figure 1.

PLOT NORMAL MOVEOUT LINES ON SEISMOGRAM OVERLAY

Tracing a seismic reflection arrival across a high-noise, large-stepout seismic record requires knowledge of the approximate normal moveouts (NMO) for each reflection. Usually one obtains these NMO's by reading their values off a normal moveout chart (Hasbrouck, 1980b), a graph on which for a specified velocity function a family of coincident-ray-time curves is plotted over a range of NMO's and source-detector distances. In our shear-wave, shallow-coal studies, however, I have found an overlay (made to the same dimensions as the seismic record) of NMO lines to be more useful in picking the records. The computer program described in this section of the report produces these overlays.

The program accepts three spread geometries: (1) a single-ended spread with equally spaced detectors, (2) a split spread with equally spaced detectors, and (3) a spread with unequally spaced detectors. If the last choice is made, then you must enter the offset distance for each trace as prompted. In the sample problems that follow, only a single-ended spread is used. Offset to the far detector, trace 1, is 66 m, detector spacing is a constant 6 m, and offset to the near detector is 0 m. Thus, the arrival time to the near detector equals the time of

the coincident ray, the ray of same down-going and up-going path.

In this program, a zero-dip reflecting horizon is assumed. Thus, the coincident ray is the vertical ray (Telford and others, 1976, p. 263). Discussion of normal moveout, NMO velocity, and construction of NOM charts is given in part 3 of the series on coal-seismic, desktop computer programs (Hasbrouck, 1980b).

Three methods of computing NMO velocity to and within the selected time interval are included in the program: (1) velocity determined from a discrete-layer model, (2) velocity derived from a downhole survey, and (3) velocity obtained from a set of X-square/T-square observations. Each of these choices is illustrated in the sample problems. If X-square/T-square velocities are selected, two methods of display of the NMO curves are offered.

After the spread information has been entered, the program prints: "ENTER RECORD WINDOW INFORMATION". You then supply the time at which the record, or record segment, to be overlain started and ended. In all the following sample problems, these times are 0 and 500, respectively. The program will also prompt: "ENTER NMO WINDOW INFORMATION". Here you are being asked to enter the coincident-ray times (TO's) for each of the NMO curves to be plotted. As an example, assume you wanted the four NMO curves with TO's of 100, 200, 300, and 400 msec. To produce these curves you would enter a start time of 100 msec, an end time of 400 msec, and an interval time of 100 msec.

Discrete four-layer model--Sample Problem 1:

Responses for this example are shown on figure 3. Tabulated near the bottom of this figure are the computed reflection times and rms velocities to the base of each of the four layers of the model followed by a listing of the NMO velocities for each of the six specified coincident-ray times. Note: for a constant velocity over the entire interval, use a one-layer model.

Figure 4 shows the six-NMO-curve, 12-trace, seismic-record overlay resulting from entry of the information on figure 3. The numbers along the top trace are the coincident-ray times for each NMO curve; the numbers beneath the bottom trace are the corresponding NMO velocities. Velocity units are the equivalents: km/sec or m/msec.

Velocity function from downhole velocity survey--Sample Problem 2:

For this example, eight NMO curves were selected, starting at TO=100 msec and ending at TO=450 msec, with an interval of 50 msec. Thus, one-way times (those recorded in a downhole velocity survey) range from 50 to 225 msec at an interval of 25 msec. Discussion of the downhole velocity survey is given in part 2 of this series of reports (Hasbrouck, 1980a). For each of the one-way times you are prompted on figure 5 to enter the downhole survey one-way times and their associated rms velocities that bracket the preselected one-way times. The program takes these data, performs a linear interpolation between the bounding times, tabulates the results (bottom of fig. 5), and then uses these values to compute the NMO curves plotted on figure 6.

SUPERPOSE NMO CURVES ON ZERO-AMPLITUDE 12-TRACE SEISMIC RECORD

ENTER RECORD AND SPREAD INFORMATION

SELECT SPREAD TYPE AND DIMENSIONS

1. Single-ended spread with equally spaced detectors
2. Split spread with equally spaced detectors
3. Spread with unequally spaced detectors

ENTER SELECTION FROM ABOVE LIST; NUMBER =

FOR THE SINGLE-ENDED SPREAD WITH EQUALLY SPACED DETECTORS:

Offset to trace-1 detector =

Offset to trace-12 detector =

ENTER RECORD WINDOW INFORMATION

Time at start of record =

Time at end of record =

SELECT VELOCITY FUNCTION TO AND WITHIN NMO INTERVAL

1. Velocity from discrete-layer model
2. Velocity from downhole velocity survey
3. Velocity from set of X₁2-T₁2 analyses

ENTER SELECTION FROM ABOVE LIST; NUMBER =

ENTER NMO WINDOW INFORMATION

Start time of NMO window =

End time of NMO window =

Interval within NMO window =

Number of NMO curves = 6

ENTER MODEL PARAMETERS

Number of layers in model =

LAYER	THICKNESS	INTERVAL VELOCITY
1	<input type="text" value="20"/>	<input type="text" value=".25"/>
2	<input type="text" value="16"/>	<input type="text" value=".4"/>
3	<input type="text" value="42"/>	<input type="text" value=".6"/>
4	<input type="text" value="10"/>	<input type="text" value=".5"/>

NOTE: VALUES TO BASE OF LAYER

LAYER	REFL TIME	RMS VEL
1	160.0	0.250
2	240.0	0.300
3	300.0	0.439
4	420.0	0.445

INTERVAL VELOCITY BENEATH LAYER 4 =

NMO VELOCITY AT SPECIFIED T0'S (COINCIDENT-RAY TIMES)

CURVE NO.	T0	NMO VEL
1	200	0.286
2	250	0.325
3	300	0.385
4	350	0.422
5	400	0.442
6	450	0.502

Figure 3. Copy of screen display showing entries for sample problem 1.




Figure 4. Copy of screen display showing results for sample problem 1.

Note on figure 6 that the NMO curve at $T_0=100$ msec crosses NMO curves at T_0 's of 150, 200, and 250 msec. One can well imagine the difficulty of interpreting a seismic record taken in such an area. Picking the records after overlaying them on NMO curves makes identification of seemingly anomalous arrivals a little more plausible.

Velocity from X-square/T-square analyses--Sample Problem 3:

Typical responses for this case are shown on figure 7. If you elect not to use the strip-display option, then answer N in response to the question, "Do you want strip display centered on T_0 ?". Once this response is given you are then asked to enter NMO window information. Here you must enter a start time of the NMO window less than the lowest coincident-ray time from the X-square/T-square data set, 150 msec in this example. Correspondingly, you must enter an end time of the NMO window such that it is greater than the largest T_0 of the data set, 350 msec in this example. After information about the survey area has been entered, you then supply the number of X-square/T-square values, followed by the ordered pairs of T_0 's and X-square/T-square velocities. Finally you enter the values of the estimated rms velocities at those times that bound the observed-data times (last two values entered on the bottom of fig. 7).

SUPERPOSE NMO CURVES ON ZERO-AMPLITUDE 12-TRACE SEISMIC RECORD

ENTER RECORD AND SPREAD INFORMATION

SELECT SPREAD TYPE AND DIMENSIONS

1. Single-ended spread with equally spaced detectors
2. Split spread with equally spaced detectors
3. Spread with unequally spaced detectors

ENTER SELECTION FROM ABOVE LIST; NUMBER =

FOR THE SINGLE-ENDED SPREAD WITH EQUALLY SPACED DETECTORS:

Offset to trace-1 detector =

Offset to trace-12 detector =

ENTER RECORD WINDOW INFORMATION

Time at start of record =

Time at end of record =

SELECT VELOCITY FUNCTION TO AND WITHIN NMO INTERVAL

1. Velocity from discrete-layer model
2. Velocity from downhole velocity survey
3. Velocity from set of X^2-T^2 analyses

ENTER SELECTION FROM ABOVE LIST; NUMBER =

ENTER NMO WINDOW INFORMATION

Start time of NMO window =

End time of NMO window =

Interval within NMO window =

Number of NMO curves = 8

ENTER DATA FROM A DOWNHOLE SURVEY

Survey area (20 char,max) :

Hole designation (7 char,max) :

Data date (12 char,max) :

CURVE NO.	1-WAY TIME, T0	BRACKETING 1-WAY TIMES SHALLOWER	DEEPER	BRACKETING VEL @ <T0	RMS VELOCITIES VEL @ >T0
1	50	45	55	.22	.28
2	75	70	80	.35	.40
3	100	92	104	.50	.52
4	125	121	140	.53	.70
5	150	140	155	.70	1.0
6	175	170	180	1.1	1.3
7	200	190	220	1.4	1.6
8	225	220	132	1.6	1.8

INTERPOLATED VEL VS COINCIDENT-RAY TIME

CURVE NO.	REFL TIME	NMO VEL
1	100.0	0.25
2	150.0	0.38
3	200.0	0.51
4	250.0	0.57
5	300.0	0.90
6	350.0	1.20
7	400.0	1.47
8	450.0	1.59

Figure 5. Copy of screen display showing entries for sample problem 2.




Figure 6. Copy of screen display showing results for sample problem 2.

Using the information entered on figure 7, the program performs a linear interpolation, computes NMO times, and makes the presentation shown on figure 8. Here the NMO curves are at equal 100-msec intervals.

X-square/T-square velocity with strip-display mode--Sample Problem 4:

To invoke strip-display mode, answer Y to the question, "Do you want strip display centered on T0?". When this option is selected you are not asked to supply NMO window information, but instead you are asked to enter the half width of the strip (10 msec in this sample problem). We can think of this as entering the beginning and end times of a 20-msec wavelet centered on the NMO curve.

To compare the displays of examples 3 and 4, we use the same X-square/T-square data as in sample problem 3. These entries are shown in the boxes on the bottom of figure 9.

SUPERPOSE NMO CURVES ON ZERO-AMPLITUDE 12-TRACE SEISMIC RECORD

ENTER RECORD AND SPREAD INFORMATION

SELECT SPREAD TYPE AND DIMENSIONS

1. Single-ended spread with equally spaced detectors
2. Split spread with equally spaced detectors
3. Spread with unequally spaced detectors

ENTER SELECTION FROM ABOVE LIST; NUMBER =

FOR THE SINGLE-ENDED SPREAD WITH EQUALLY SPACED DETECTORS:

Offset to trace-1 detector =

Offset to trace-12 detector =

ENTER RECORD WINDOW INFORMATION

Time at start of record =

Time at end of record =

SELECT VELOCITY FUNCTION TO AND WITHIN NMO INTERVAL

1. Velocity from discrete-layer model
2. Velocity from downhole velocity survey
3. Velocity from set of X²-T² analyses

ENTER SELECTION FROM ABOVE LIST; NUMBER =

Do you want strip display centered on T₀? (Y OR N)

ENTER NMO WINDOW INFORMATION

Start time of NMO window =

End time of NMO window =

Interval within NMO window =

Number of NMO curves =

Survey area (20 char,max) :

SP designation (7 char,max) :

Data date (12 char,max) :

Number of X²-T² values =

COINCIDENT-RAY TIME

X²-T² VEL

Est. RMS vel for first NMO curve (T₀=50) =

Est. RMS vel for last NMO curve (T₀=450) =

Figure 7. Copy of screen display showing entries for sample problem 3.

The result of these inputs are displayed on figure 10. In making this presentation, no interpolations are called for; thus, we see only those NMO curves using velocities determined from the X-square/T-square analyses. Note the compression of wave forms as a function of normal moveout; compare, for example, width of strips for T₀'s of 50, 150, and 250 msec.




Figure 8. Copy of screen display showing results of sample problem 3.

COMMENTS ON AND LISTINGS OF THE PROGRAMS

In order to put the programs of this report to work, you must know how to perform the following operations:

1. transcribe the programs into the computer,
2. store the programs on magnetic tape,
3. retrieve the programs from magnetic tape,
4. enter information from the keyboard, and
5. copy the screen display.

These tasks are well documented in the computer's operator's manuals.

Four control characters (ones requiring the holding down of the control key as the letter is entered) are used in the programs: G (ring bell), K (move cursor up one line), L (erase screen and move cursor to the HOME position), and the RUB OUT (move cursor to the left margin and down one line). In the printed listing these control characters are shown as G_, K_, L_, and _.

If your computer does not contain a clock, you must either replace the time statements (for example, CALL "TIME",A\$) by a REMARK statement and then remove all references to time-term string variables in various

SUPERPOSE NMO CURVES ON ZERO-AMPLITUDE 12-TRACE SEISMIC RECORD

ENTER RECORD AND SPREAD INFORMATION

SELECT SPREAD TYPE AND DIMENSIONS

1. Single-ended spread with equally spaced detectors
2. Split spread with equally spaced detectors
3. Spread with unequally spaced detectors

ENTER SELECTION FROM ABOVE LIST; NUMBER =

FOR THE SINGLE-ENDED SPREAD WITH EQUALLY SPACED DETECTORS:

Offset to trace-1 detector =

Offset to trace-12 detector =

ENTER RECORD WINDOW INFORMATION

Time at start of record =

Time at end of record =

SELECT VELOCITY FUNCTION TO AND WITHIN NMO INTERVAL

1. Velocity from discrete-layer model
2. Velocity from downhole velocity survey
3. Velocity from set of $X\uparrow 2-T\uparrow 2$ analyses

ENTER SELECTION FROM ABOVE LIST; NUMBER =

Do you want strip display centered on T0? (Y OR N)

Half width of strip in msec =

Survey area (20 char,max) :

SP designation (7 char,max) :

Data date (12 char,max) :

Number of $X\uparrow 2-T\uparrow 2$ values =

COINCIDENT-RAY TIME

$X\uparrow 2-T\uparrow 2$ VEL

Figure 9. Copy of screen display showing entries for sample problem 4.

print statements, or you must enter the appropriate times after adding prompts such as illustrated below:

```
1000 CALL "TIME",A$
```

```
2000 CALL "TIME",B$
```

replaced by:

```
1000 PRINT "START TIME = ";
```

```
1002 INPUT A$
```

```
2000 PRINT "END TIME = ";
```

```
2002 INPUT B$
```




Figure 10. Copy of screen display showing results for sample problem 4.

REFERENCES

- Dix, C. H., 1955, Seismic velocities from surface measurements: Geophysics, v. 20, no. 1, p. 68-86.
- Green, C. H., 1938, Velocity determinations by means of reflection profiles: Geophysics, v. 3, no. 4, p. 295-348.
- Hasbrouck, W. P., 1980a, Coal-seismic, desk-top computer programs in BASIC; part 2: enter, compute, display, edit, and store results of downhole, inhole, and crosshole investigations: U.S. Geological Survey Open-File Report 80-669, 88 p.
- Hasbrouck, W. P., 1980b, Coal-seismic, desk-top computer programs in BASIC; part 3: compute, tabulate, and plot normal moveout times: U.S. Geological Survey Open-File Report 80-670, 20 p.
- Hasbrouck, W. P. and Padgett, Nigel, 1982, Use of shear-wave seismics in the evaluation of strippable coal resources, in K. D. Gurgel, ed., Proceedings, fifth symposium on the geology of Rocky Mountain coal: Utah Geological and Mineral Survey, Bull 118, p. 203-210.
- Telford, W. M., Geldart, L. P., Sheriff, R. E., and Keys, D. A., 1976, Applied Geophysics: New York, Cambridge University Press, 860 p.

LISTING OF THE X-SQUARE/ T-SQUARE PROGRAM

```

100 PRINT "L_X-SQUARE/ T-SQUARE ANALYSIS PROGRAM"
110 INIT
120 DIM A$(20), B$(9), C$(9), D$(18), E$(1), G$(1), H$(10), L$(12), O$(13)
130 DIM S$(18), X$(12), Y$(20), Z$(13)
140 B$="T SQUARED"
150 C$="X SQUARED"
160 X$="X2 T2 VEL="
170 Y$="COINCIDENT-RAY TIME="
180 Z$="NORMAL DEPTH="
190 REM *** ENTER RECORD AND LOCATION INFORMATION
200 GOSUB 520
210 REM *** ENTER DETECTOR DISTANCES AND REFLECTION TIMES
220 GOSUB 660
230 MOVE 0,0
240 PRINT
250 REM *** PLOT X-SQUARE/ T-SQUARE DATA
260 GOSUB 880
270 REM *** COMPUTE X-SQUARE/ T-SQUARE VELOCITY
280 GOSUB 1200
290 REM *** PLOT LEAST SQUARE LINE
300 MOVE X2(1), Y1
310 DRAW X2(N1), Y2
320 REM *** PRINT VELOCITY, COINCIDENT-RAY TIME, AND NORMAL DEPTH AT X=0
330 WINDOW 0,130,0,100
340 VIEWPORT 0,130,0,100
350 IMAGE 12A,D,3D," km/sec"
360 MOVE 0,4
370 PRINT USING 350:X$,Y1
380 IMAGE 26X,20A,4D
390 MOVE 0,4
400 PRINT USING 380:Y$,I2
410 IMAGE 52X,13A,3D,D
420 MOVE 0,4
430 PRINT USING 410:Z$,Z0
440 MOVE B1,B5
450 REM *** COMPUTE AND TABULATE TIME DEVIATIONS
460 GOSUB 1370
470 GOSUB 1440
480 MOVE 0,0
490 PRINT
500 PRINT "G_G_G___PROGRAM COMPLETED"
510 END
520 REM *** SUB:ENTER RECORD AND LOCATION INFORMATION
530 PRINT "___ HEADER ON RECORD (8 characters) = ";
540 INPUT H$
550 PRINT "   AREA NAME (20 characters,max) = ";
560 INPUT A$
570 PRINT "   LATITUDE (12 characters,max) = ";
580 INPUT L$
590 PRINT "   LONGITUDE (13 characters,max) = ";

```

```

600 INPUT O$
610 PRINT " SOURCE TYPE (18 characters,max) = ";
620 INPUT S$
630 PRINT "DETECTOR TYPE (18 characters,max) = ";
640 INPUT D$
650 RETURN
660 REM ** SUB:ENTER DETECTOR DISTANCES AND REFLECTION TIMES
670 PRINT " DISTANCE TO FIRST DETECTOR = ";
680 INPUT D1
690 PRINT " DISTANCE TO LAST DETECTOR = ";
700 INPUT D2
710 PRINT " DETECTOR INTERVAL = ";
720 INPUT D3
730 N1=(D2-D1)/D3+1
740 DIM T1(N1),T2(N1),T3(N1),T4(N1),X1(N1),X2(N1)
750 PRINT "___INDEX NO. DETECTOR DISTANCE REFLECTION TIME"
760 D4=D1-D3
770 FOR J=1 TO N1
780 D4=D4+D3
790 X1(J)=D4
800 PRINT " ";J;" ";X1(J); " ";
810 INPUT T1(J)
820 NEXT J
830 CALL "MIN",T1,M1,I1
840 CALL "MAX",T1,M2,I2
850 X2=X1+2
860 T2=T1+2
870 RETURN
880 REM ** SUB: PLOT X-SQUARE, T-SQUARE DATA
890 REM ** PRINT TOP LABELS
900 PRINT H$;" AREA: ";A$;" LAT: ";L$;" LONG: ";O$
910 PRINT "SOURCE TYPE: ";S$;" DETECTOR TYPE: ";D$
920 PRINT "MIN X: ";D1;" MAX X: ";D2;" DELTA X: ";D3;" ";
930 PRINT " MIN T: ";M1;" MAX T: ";M2
940 RESTORE 950
950 DATA 5,129,10,89,1,806,2,957,0
960 READ B1,B2,B4,B5,K1,K2,V1
970 B3=B2-B1
980 B6=B5-B4
990 REM ** PLOT BORDER
1000 MOVE B1,B5
1010 RDRAW B3,0
1020 RDRAW 0,-B6
1030 RDRAW -B3,0
1040 RDRAW 0,B6
1050 MOVE 0,(B4+B5)/2+4.5*K2
1060 FOR J=1 TO 9
1070 E$=SEG(B$,J,1)
1080 PRINT E$
1090 NEXT J

```

```

1100 MOVE (B1+B2)/2-4.5*K1,7
1110 PRINT C$
1120 REM ** PLOT X,T,T2
1130 WINDOW X2(1),X2(N1),M1*M1,M2*M2
1140 VIEWPORT B1,B2,B4,B5
1150 MOVE 0,0
1160 FOR J=1 TO N1
1170 DRAW X2(J),T2(J)
1180 NEXT J
1190 RETURN
1200 REM ** SUB: COMPUTE X-SQUARE, T-SQUARE VELOCITY
1210 A1=SUM(X2)/N1
1220 A2=SUM(T2)/N1
1230 A3=0
1240 A4=0
1250 FOR J=1 TO N1
1260 A3=A3+X2(J)*X2(J)
1270 A4=A4+X2(J)*T2(J)
1280 NEXT J
1290 S1=(A4-N1*A1*A2)/(A3-N1*A1*A1)
1300 V1=SQR(1/S1)
1310 I1=A2-S1*A1
1320 I2=SQR(I1)
1330 Z0=0.5*I2*V1
1340 Y1=S1*X2(1)+I1
1350 Y2=S1*X2(N1)+I1
1360 RETURN
1370 REM ** SUB: COMPUTE TIME DEVIATIONS FROM LEAST SQUARE LINE
1380 FOR J=1 TO N1
1390 T3(J)=S1*X2(J)+I1
1400 NEXT J
1410 T3=SQR(T3)
1420 T4=T1-T3
1430 RETURN
1440 REM ** SUB: TABULATE TIME DEVIATIONS (STATICS)
1450 T4=-T4
1460 PRINT " DIST STATIC CORR"
1470 IMAGE 3X,3D,D,4X,3D,D
1480 N2=0
1490 FOR J=1 TO N1
1500 N2=N2+1
1510 IF N2>20 THEN 1550
1520 PRINT USING 1470:X1(J),T4(J)
1530 NEXT J
1540 RETURN
1550 MOVE B1,B5
1560 PRINT " DIST STATIC CORR"
1570 IMAGE 24X,3D,D,4X,3D,D
1580 FOR J=29 TO N1
1590 PRINT USING 1570:X1(J),T4(J)
1600 NEXT J
1610 RETURN

```


LISTING OF NMO-OVERLAY PROGRAM

```

100 PRINT "L-SUPERPOSE NMO CURVES ON ZERO-AMPLITUDE";
110 PRINT " 12-TRACE SEISMIC RECORD"
120 INIT
130 DIM B$(18), G$(1), X1(12)
140 Q1=1
150 CALL "TIME", B$
160 PRINT "___ENTER RECORD AND SPREAD INFORMATION___"
170 PRINT "SELECT SPREAD TYPE AND DIMENSIONS"
180 PRINT " 1. Single-ended spread with equally spaced detectors"
190 PRINT " 2. Split spread with equally spaced detectors"
200 PRINT " 3. Spread with unequally spaced detectors"
210 PRINT "ENTER SELECTION FROM ABOVE LIST; NUMBER = ";
220 INPUT N0
230 GO TO N0 OF 240, 260, 280
240 GOSUB 710
250 GO TO 290
260 GOSUB 820
270 GO TO 290
280 GOSUB 1010
290 PRINT "___ENTER RECORD WINDOW INFORMATION"
300 PRINT "      Time at start of record = ";
310 INPUT T1
320 PRINT "      Time at end of record = ";
330 INPUT T2
340 PRINT "___SELECT VELOCITY FUNCTION TO AND WITHIN NMO INTERVAL"
350 PRINT " 1. Velocity from discrete-layer model"
360 PRINT " 2. Velocity from downhole velocity survey"
370 PRINT " 3. Velocity from set of Xt2-Tt2 analyses"
380 PRINT "ENTER SELECTION FROM ABOVE LIST; NUMBER = ";
390 INPUT N0
400 GO TO N0 OF 410, 440, 470
410 GOSUB 1160
420 GOSUB 1340
430 GO TO 570
440 GOSUB 1160
450 GOSUB 1980
460 GO TO 570
470 Q1=1
480 PRINT " Do you want strip display centered on T0? (Y OR N) ";
490 INPUT G$
500 IF G$="Y" THEN 530
510 GOSUB 1160
520 GO TO 560
530 Q1=2
540 PRINT "      Half width of strip in msec = ";
550 INPUT T3
560 GOSUB 2280
570 REM *** COMPUTE NMO-TIMES
580 GOSUB 2920
590 REM *** PRINT LABEL AND PLOT RECORD BORDER AND TIME LINES
600 GOSUB 2990
610 REM *** PLOT ZERO-AMPLITUDE TRACES
620 GOSUB 3620
630 REM *** PLOT NMO CURVES
640 IF Q1=2 THEN 670
650 GOSUB 3760
660 GO TO 680
670 GOSUB 3900
680 GOSUB 1100
690 PRINT "G_G_G___PROGRAM COMPLETED"
700 END
710 REM *** SUB: SINGLE-ENDED SPREAD WITH EQUALLY SPACED DETECTORS
720 PRINT "___FOR THE SINGLE-ENDED SPREAD WITH EQUALLY SPACED DETECTORS:"
730 PRINT "Offset to trace-1 detector = ";
740 INPUT X1(1)
750 PRINT "Offset to trace-12 detector = ";

```

```

760 INPUT X1(12)
770 X2=(X1(12)-X1(1))/11
780 FOR J=2 TO 11
790 X1(J)=X1(J-1)+X2
800 NEXT J
810 RETURN
820 REM *** SUB: SPLIT SPREAD WITH EQUALLY SPACED DETECTORS
830 PRINT "___FOR THE SPLIT SPREAD WITH EQUALLY SPACED DETECTORS:"
840 PRINT "Offset to trace-1 detector = ";
850 INPUT X1(1)
860 PRINT "Offset to trace-6 detector = ";
870 INPUT X1(6)
880 X2=(X1(6)-X1(1))/5
890 PRINT "Offset to trace-7 detector = ";
900 INPUT X1(7)
910 PRINT "Offset to trace-12 detector = ";
920 INPUT X1(12)
930 X3=(X1(12)-X1(7))/5
940 FOR J=2 TO 5
950 X1(J)=X1(J-1)+X2
960 NEXT J
970 FOR J=9 TO 11
980 X1(J)=X1(J-1)+X3
990 NEXT J
1000 RETURN
1010 REM *** SUB: SPREAD WITH UNEQUALLY SPACED DETECTORS
1020 PRINT "FOR THE SPREAD WITH UNEQUALLY SPACED DETECTORS:"
1030 PRINT "TRACE NO. OFFSET"
1040 FOR J=1 TO 12
1050 PRINT " ", J, " ";
1060 INPUT X1(J)
1070 NEXT J
1080 GOSUB 1100
1090 RETURN
1100 REM *** SUB: DEFAULT DISPLAY AND MOVE TO PAGE BOTTOM
1110 WINDOW 0,130,0,100
1120 VIEWPORT 0,130,0,100
1130 MOVE 0,0
1140 PRINT
1150 RETURN
1160 REM *** SUB: ENTER NMO WINDOW INFORMATION
1170 PRINT "___ENTER NMO WINDOW INFORMATION"
1180 PRINT "      Start time of NMO window = ";
1190 INPUT T8
1200 PRINT "      End time of NOM window = ";
1210 INPUT T9
1220 PRINT "      Interval within NMO window = ";
1230 INPUT T4
1240 N=(T9-T8)/T4+1
1250 PRINT "      Number of NMO curves = "; N
1260 DIM T0(N), T(N,12), V2(N)
1270 T0(1)=T8
1280 T0(N)=T9
1290 DELETE T8, T9
1300 FOR J=2 TO N-1
1310 T0(J)=T0(J-1)+T4
1320 NEXT J
1330 RETURN
1340 REM *** SUB: COMPUTE VELOCITIES FROM MODEL
1350 Q=1
1360 GOSUB 1100
1370 PRINT "ENTER MODEL PARAMETERS"
1380 PRINT "      Number of layers in model = ";
1390 INPUT N1
1400 N3=N MAX N1
1410 DIM B(N3), C3(N1), T3(N1), V0(N1+1), V3(N3), V4(N3), Z3(N1)
1420 A1=0
1430 B1=0

```

```

1440 C1=0
1450 PRINT "LAYER THICKNESS INTERVAL VELOCITY"
1460 FOR J=1 TO N1
1470 PRINT " "; J; " ";
1480 INPUT Z0
1490 PRINT "K_ ";
1500 INPUT V0(J)
1510 B(J)=Z0/V0(J)
1520 B1=B1+B(J)
1530 T3(J)=2*B1
1540 C1=C1+V0(J)*V0(J)*B(J)
1550 C3(J)=C1
1560 V4(J)=SQR(C1/B1)
1570 NEXT J
1580 IMAGE 2X, 2D, 7X, 4D, D, 7X, 2D, 3D
1590 PRINT "___NOTE: VALUES TO BASE OF LAYER"
1600 PRINT "LAYER REFL TIME RMS VEL"
1610 FOR J=1 TO N1
1620 PRINT USING 1580: J, T3(J), V4(J)
1630 NEXT J
1640 PRINT "___INTERVAL VELOCITY BENEATH LAYER "; N1; " = ";
1650 INPUT V0(N1+1)
1660 REM *** DETERMINE NMO VELOCITY AT SPECIFIED T0(J) TIME
1670 FOR J=1 TO N
1680 IF T0(J)>T3(1) THEN 1710
1690 V2(J)=V0(1)
1700 GO TO 1790
1710 FOR L=1 TO N1
1720 IF T0(J)>T3(L) THEN 1740
1730 GO TO 1790
1740 B1=0.5*(T0(J)-T3(L))
1750 B2=0.5*T0(J)
1760 C2=C3(L)+V0(L+1)*V0(L+1)*B1
1770 V2(J)=SQR(C2/B2)
1780 NEXT L
1790 NEXT J
1800 PRINT "___NMO VELOCITY AT SPECIFIED T0'S (COINCIDENT-RAY TIMES)"
1810 PRINT "CURVE NO. T0 NMO VEL"
1820 IMAGE 3X, 2D, 6X, 4D, 4X, 2D, 3D
1830 FOR J=1 TO N
1840 PRINT USING 1820: J, T0(J), V2(J)
1850 NEXT J
1860 DELETE B, B1, B2, C1, C2, C3, T3, V4
1870 RETURN
1880 REM *** SUB: COMPUTE VELOCITIES FROM DOWNHOLE-SURVEY RESULTS
1890 GOSUB 1100
1900 Q=2
1910 DIM A$(20), U$(7), D$(12)
1920 PRINT "ENTER DATA FROM A DOWNHOLE SURVEY"
1930 PRINT " Survey area (20 char,max) : ";
1940 INPUT A$
1950 PRINT "Hole designation (7 char,max) : ";
1960 INPUT U$
1970 PRINT " Data date (12 char,max) : ";
1980 INPUT D$
1990 T0=0.5*T0
2000 PRINT "___CURVE 1-WAY BRACKETING 1-WAY TIMES";
2010 PRINT " BRACKETING RMS VELOCITIES"
2020 PRINT " NO. TIME T0 SHALLOWER DEEPER VEL @ <T0";
2030 PRINT " VEL @ >T0 "
2040 FOR J=1 TO N
2050 PRINT " "; J; " "; T0(J)
2060 PRINT "K_ ";
2070 INPUT T5
2080 PRINT "K_ ";
2090 INPUT T6
2100 K=(T0(J)-T5)/(T6-T5)
2110 PRINT "K_ ";

```

```

2120 INPUT V7
2130 PRI "K_
2140 INPUT V8
2150 V2(J)=V7+K*(V8-V7)
2160 NEXT J
2170 T0=2*T0
2180 GOSUB 1100
2190 PRINT "INTERPOLATED VEL VS COINCIDENT-RAY TIME"
2200 PRINT "CURVE NO. REFL TIME NMO VEL"
2210 IMAGE 3X, 2D, 8X, 4D, D, 8X, 2D, 2D
2220 FOR J=1 TO N
2230 PRINT USING 2210: J, T0(J), V2(J)
2240 NEXT J
2250 Q=2
2260 GOSUB 1100
2270 RETURN
2280 REM *** SUB: COMPUTE VELOCITIES FROM SET OF X+2-T+2 ANALYSES
2290 GOSUB 1100
2300 Q=3
2310 DIM A$(20), U$(70), D$(12)
2320 PRINT " Survey area (20 char,max) : ";
2330 INPUT A$
2340 PRINT "SP designation (7 char,max) : ";
2350 INPUT U$
2360 PRINT " Data date (12 char,max) : ";
2370 INPUT D$
2380 PRINT "___ Number of X+2-T+2 values = ";
2390 INPUT N1
2400 N1=2+N1
2410 DIM T3(N1), V3(N1)
2420 PRINT "___COINCIDENT-RAY TIME X+2-T+2 VEL"
2430 FOR J=2 TO N1-1
2440 PRINT " ";
2450 INPUT T3(J)
2460 PRINT "K_ ";
2470 INPUT V3(J)
2480 NEXT J
2490 IF Q1=2 THEN 2680
2500 REM *** LINEAR INTERPOLATION OF INTERMEDIATE RMS VALUES
2510 PRINT "___Est. RMS vel for first NMO curve (T0="; T0(1); ") = ";
2520 INPUT V3(1)
2530 V2(1)=V3(1)
2540 PRINT "Est. RMS vel for last NMO curve (T0="; T0(N); ") = ";
2550 INPUT V3(N1)
2560 V2(N)=V3(N1)
2570 T3(1)=T0(1)
2580 T3(N1)=T0(N)
2590 FOR J=2 TO N-1
2600 FOR K=1 TO N1-1
2610 IF T0(J)>T3(K) AND T0(J)<T3(K+1) THEN 2640
2620 NEXT K
2630 GO TO 2670
2640 C=(V3(K+1)-V3(K))/(T3(K+1)-T3(K))
2650 V2(J)=C*(T0(J)-T3(K))+V3(K)
2660 NEXT J
2670 RETURN
2680 REM *** STRIP DISPLAY
2690 N=3*(N1-2)
2700 DIM T0(N), T(N, 12), V2(N)
2710 K=1
2720 FOR J=2 TO N STEP 3
2730 K=K+1
2740 T0(J-1)=T3(K)-T5
2750 V2(J-1)=V3(K)
2760 T0(J)=T3(K)
2770 V2(J)=V3(K)
2780 T0(J+1)=T3(K)+T5
2790 V2(J+1)=V3(K)

```

```

2800 NEXT J
2810 RETURN
2820 REM ** SUB: COMPUTE NMO TIMES
2830 FOR J=1 TO N
2840 FOR K=1 TO 12
2850 T(J,K)=SQR(T0(J)*T0(J)+X1(K)/V2(J)*X1(K)/V2(J))-T0(J)
2860 NEXT K
2870 NEXT J
2880 RETURN
2890 REM ** SUB: PRINT LABEL AND PLOT RECORD BORDER AND TIME LINES
2900 C2=130
2910 D2=92
2920 D3=02/13
2930 D4=02
2940 GOSUB 1100
2950 B$=SEG(B$,1,9)
2960 GO TO 0 OF 2970,3000,3040
2970 PRINT "L_NMO CURVES USING RESULTS FROM ";N1;"-LAYER MODEL";
2980 PRINT "      COMP. DATE: ";B$
2990 GO TO 3080
3000 PRINT "L_NMO CURVES USING RESULTS FROM DOWNHOLE VELOCITY SURVEY";
3010 PRINT "      DATE: ";B$
3020 PRINT "AREA: ";A$;"      HOLE: ";U$;"      DATA TAKEN ON ";D$
3030 GO TO 3080
3040 PRINT "L_NMO CURVES USING VELOCITIES FROM X+2-T+2 ANALYSES";
3050 PRINT "      COMP. DATE: ";B$
3060 PRINT "AREA: ";A$;"      SP: ";U$;"      DATA TAKEN ON ";D$
3070 REM
3080 MOVE 0,D2+0.5
3090 PRINT "O/S"
3100 IMAGE 3D,0
3110 MOVE C2/10-2.34,D2+0.5
3120 T7=(T2-T1)/10
3130 PRINT USING 3100:T1+T7
3140 FOR K=2 TO 9
3150 RMOVE C2/10,0
3160 PRINT USING 3100:T1+K*T7
3170 NEXT K
3180 REM ** PLOT RECORD BORDER AND TIME LINES
3190 C3=C2/50
3200 DIM B(300)
3210 B=02
3220 M1=0.4
3230 M2=48
3240 M3=02-M1
3250 M4=02-M2
3260 B(1)=0
3270 C4=0
3280 FOR K=3 TO 291 STEP 6
3290 C4=C4+C3
3300 B(K)=C4
3310 B(K+2)=C4
3320 B(K+3)=M3
3330 B(K+4)=C4
3340 NEXT K
3350 B(297)=130
3360 B(299)=130
3370 B(300)=0
3380 FOR K=30 TO 270 STEP 30
3390 B(K)=M4
3400 MOVE 0,D2
3410 NEXT K
3420 PRINT @32,20:B
3430 DELETE B
3440 DIM B(300)
3450 B=0
3460 B(1)=130
3470 C4=130

```

```

3480 FOR K=3 TO 291 STEP 6
3490 C4=C4-C3
3500 B(K)=C4
3510 B(K+2)=C4
3520 B(K+3)=M1
3530 B(K+4)=C4
3540 NEXT K
3550 FOR K=30 TO 270 STEP 30
3560 B(K)=M2
3570 NEXT K
3580 B(300)=D2
3590 PRINT @32, 20:B
3600 DELETE B
3610 RETURN
3620 REM *** SUB: PLOT ZERO-AMPLITUDE TRACES
3630 MOVE 1, D2-D3+3
3640 PRINT "T0"
3650 MOVE 0, D2
3660 FOR K=1 TO 12
3670 D4=D4-D3
3680 MOVE 0, D4
3690 PRINT X1(K)
3700 MOVE 0, D4
3710 RDRAW C2, 0
3720 NEXT K
3730 MOVE 1, D4-3
3740 PRINT "VEL"
3750 RETURN
3760 REM *** SUB: PLOT NMO'S USING LINE DISPLAY
3770 WINDOW T1, T2, 0, 100
3780 FOR J=1 TO N
3790 MOVE T(J, 1)+T0(J), D2-D3
3800 IMAGE "H_L", 40
3810 PRINT USING 3800:T0(J)
3820 MOVE T(J, 1)+T0(J), D2-D3
3830 FOR K=2 TO 12
3840 RDRAW T(J, K)-T(J, K-1), -D3
3850 NEXT K
3860 IMAGE "H_J", 0, 20
3870 PRINT USING 3860:V2(J)
3880 NEXT J
3890 RETURN
3900 REM *** SUB: PLOT NMO'S USING STRIP DISPLAY
3910 WINDOW T1, T2, 0, 100
3920 FOR J=2 TO N STEP 3
3930 MOVE T(J, 1)+T0(J), D2-D3
3940 PRINT "H_"; T0(J)
3950 MOVE T(J, 1)+T0(J), D2-D3
3960 FOR K=2 TO 12
3970 RDRAW T(J, K)-T(J, K-1), -D3
3980 NEXT K
3990 PRINT "H_J_"; V2(J)
4000 NEXT J
4010 FOR L=1 TO 3 STEP 2
4020 FOR J=L TO N STEP 3
4030 MOVE T(J, 1)+T0(J), D2-D3
4040 FOR K=2 TO 12
4050 RDRAW 0.125*(T(J, K)-T(J, K-1)), -0.125*D3
4060 RMOVE 0.25*(T(J, K)-T(J, K-1)), -0.25*D3
4070 RDRAW 0.25*(T(J, K)-T(J, K-1)), -0.25*D3
4080 RMOVE 0.25*(T(J, K)-T(J, K-1)), -0.25*D3
4090 RDRAW 0.125*(T(J, K)-T(J, K-1)), -0.125*D3
4100 NEXT K
4110 NEXT J
4120 NEXT L
4130 RETURN

```